Pensieve header: Analyzing Reidemeister moves by hand.

Integration

As in Talks/Beijing-2407

In[∘]:=
```
CF[ω_. ℰ_𝔼] := CF[ω] CF /@ ℰ;
CF[ℰ_List] := CF /@ ℰ;
CF[q_] := Expand@Collect[q, ϵ | (p | x | π | ξ)_, F] /. F → Factor;
```

In[∘]:=
```
𝔼 /: 𝔼[A_] 𝔼[B_] := 𝔼[A + B];
```

In[∘]:=
```
$π = Identity;
```

In[∘]:=
```
Unprotect[Integrate];
∫ω_. 𝔼[L_] ⅆ(vs_List) := Module[{n, L0, Q, Δ, G, Z0, Z, λ, DZ, DDZ, FZ, a, b},
   n = Length@vs; L0 = L /. ϵ → 0;
   Q = Table[(-∂_{vs⟦a⟧,vs⟦b⟧} L0) /. Thread[vs → 0] /. (p | x)__ → 0, {a, n}, {b, n}];
   If[(Δ = Det[Q]) == 0, Return@"Degenerate Q!"];
   Z = Z0 = CF@$π[L + vs.Q.vs / 2]; G = Inverse[Q];
   FixedPoint[(DZ = Table[∂_v Z, {v, vs}];
      DDZ = Table[∂_u DZ, {u, vs}];
      FZ = Sum[G⟦a, b⟧ (DDZ⟦a, b⟧ + DZ⟦a⟧ DZ⟦b⟧), {a, n}, {b, n}] / 2;
      Z = CF[Z0 + ∫_0^λ $π[FZ] ⅆλ]) &, Z];
   PowerExpand@Factor[ω Δ^(-1/2)] 𝔼[CF[Z /. λ → 1 /. Thread[vs → 0]]]];
Protect[Integrate];
```

In[∘]:=
```
∫𝔼[-μ x^2 / 2 + ⅈ ξ x] ⅆ{x}
```

Out[∘]=

$$\frac{\mathbb{E}\left[-\frac{\xi^2}{2\mu}\right]}{\sqrt{\mu}}$$

In[∘]:=
```
FofG = ∫𝔼[-μ (x - a)^2 / 2 + ⅈ ξ x] ⅆ{x}
```

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{i\,(2\,a\,\mu+i\,\xi)\,\xi}{2\,\mu}\right]}{\sqrt{\mu}}$$

In[∘]:=
```
∫FofG 𝔼[-𝕚 ξ x] ⅆ{ξ}
```

Out[∘]=

$$\mathbb{E}\left[-\frac{1}{2}\,(a-x)^2\,\mu\right]$$

In[∘]:=
```
L = -1/2 {x1, x2}.(a b
                   b c).{x1, x2} + {ξ1, ξ2}.{x1, x2};
Z12 = ∫𝔼[L] ⅆ{x1, x2}
```

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{c\,\xi_1^2}{2\,(-b^2+a\,c)}+\frac{b\,\xi_1\,\xi_2}{b^2-a\,c}+\frac{a\,\xi_2^2}{2\,(-b^2+a\,c)}\right]}{\sqrt{-b^2+a\,c}}$$

In[∘]:=
```
{Z1 = ∫𝔼[L] ⅆ{x1}, Z12 == ∫Z1 ⅆ{x2}}
```

Out[∘]=

$$\left\{\frac{\mathbb{E}\left[-\frac{(-b^2+a\,c)\,x_2^2}{2\,a}-\frac{b\,x_2\,\xi_1}{a}+\frac{\xi_1^2}{2\,a}+x_2\,\xi_2\right]}{\sqrt{a}},\ \text{True}\right\}$$

In[∘]:=
```
$π = Normal[# + O[ϵ]^13] &; ∫𝔼[-ϕ^2/2 + ϵ ϕ^3/6] ⅆ{ϕ}
```

Out[∘]=

$$\mathbb{E}\left[\frac{5\,\epsilon^2}{24}+\frac{5\,\epsilon^4}{16}+\frac{1105\,\epsilon^6}{1152}+\frac{565\,\epsilon^8}{128}+\frac{82\,825\,\epsilon^{10}}{3072}+\frac{19\,675\,\epsilon^{12}}{96}\right]$$

The Quadratic

Matrices from SSAlexander4Tangles.nb

In[∘]:=



$$\left(\begin{array}{l} M=\begin{pmatrix} -1 & -1 & 2 & 0\\ 0 & 0 & 0 & 0\\ 0 & 1 & -1 & 0\\ 1 & 0 & -1 & 0\end{pmatrix}\\ \overline{M}=\begin{pmatrix} 1 & -1 & 0 & 0\\ 0 & 0 & 0 & 0\\ -2 & 1 & 1 & 0\\ 1 & 0 & -1 & 0\end{pmatrix}\end{array}\right);$$

In[∘]:=
```
Q_{i_,j_,k_,l_} := {p_i, p_j, p_k, p_l}.(t M - t^-1 M^T).{x_i, x_j, x_k, x_l};
Q̄_{i_,j_,k_,l_} := {p_i, p_j, p_k, p_l}.(t M̄ - t^-1 M̄^T).{x_i, x_j, x_k, x_l};
```

In[∘]:=
```
Simplify[Q_{i,j,k,l}]
```

Out[∘]=

$$\frac{1}{t}\left(\left(-\left(\left(-1+t^2\right)p_i\right)+p_j-2\,p_k+t^2\,p_l\right)x_i-t^2\,(p_i-p_k)\,x_j+\left(2\,t^2\,p_i-p_j+p_k-t^2\,p_k-t^2\,p_l\right)x_k-(p_i-p_k)\,x_l\right)$$

R2b



In[∘]:= $Q_{1,2,3,4}+\overline{Q}_{3,2,5,4}$ // Expand

Out[∘]=

$$\frac{p_1 x_1}{t}-t\,p_1 x_1+\frac{p_2 x_1}{t}-\frac{2 p_3 x_1}{t}+t\,p_4 x_1-t\,p_1 x_2+t\,p_5 x_2+2\,t\,p_1 x_3-2\,t\,p_5 x_3-\frac{p_1 x_4}{t}+\frac{p_5 x_4}{t}-\frac{p_2 x_5}{t}+\frac{2 p_3 x_5}{t}-t\,p_4 x_5-\frac{p_5 x_5}{t}+t\,p_5 x_5$$

In[∘]:= Expand@Collect$\left[Q_{1,2,3,4}+\overline{Q}_{3,2,5,4},\ (p\mid x)_3,\ F\right]$ /. F → Factor

Out[∘]=

$$2\,t\,(p_1-p_5)\,x_3-\frac{2 p_3 (x_1-x_5)}{t}-\frac{1}{t}\left(-p_1 x_1+t^2 p_1 x_1-p_2 x_1-t^2 p_4 x_1+t^2 p_1 x_2-t^2 p_5 x_2+p_1 x_4-p_5 x_4+p_2 x_5+t^2 p_4 x_5+p_5 x_5-t^2 p_5 x_5\right)$$

In[∘]:= Expand$\left[Q_{1,2,3,4}+\overline{Q}_{3,2,5,4}\right.$ /. $\{p_5\to p_1,\ x_5\to x_1\}\left.\right]$

Out[∘]=

$$0$$

R3



In[∘]:= lhs = Simplify$[Q_{1,2,7,6}+Q_{2,3,4,7}+Q_{7,4,5,6}]$;
rhs = Simplify$[Q_{2,3,7,1}+Q_{1,7,5,6}+Q_{7,3,4,5}]$

Out[∘]=

$$\frac{1}{t}\Big(-2 p_5 x_1+t^2 p_6 x_1+2 p_7 x_1+p_3 x_2-2 p_7 x_2+t^2 p_4 x_3-p_3 x_4+p_4 x_4-t^2 p_4 x_4-t^2 p_5 x_4+2 t^2 p_7 x_4+p_4 x_5+p_5 x_5-t^2 p_5 x_5-t^2 p_6 x_5-2 p_7 x_5+p_5 x_6-p_2\left(x_1+\left(-1+t^2\right) x_2+t^2 (x_3-2 x_7)\right)-2 p_4 x_7+2 t^2 p_5 x_7+2 p_7 x_7-2 t^2 p_7 x_7+p_1\left(-\left(\left(-1+t^2\right) x_1\right)+t^2 x_2+2 t^2 x_5-x_6-2 t^2 x_7\right)\Big)$$

In[∘]:= `Simplify[lhs == rhs]`

Out[∘]=

$$\frac{1}{t}\left(-2 p_7 x_1 - t^2 p_1 x_2 - p_4 x_2 + p_7 x_2 + t^2 p_7 x_2 - 2 t^2 p_7 x_4 - t^2 p_1 x_5 - p_4 x_5 + p_7 x_5 + t^2 p_7 x_5 + 2 t^2 p_1 x_7 + 2 p_4 x_7 + p_2 \left(x_1 + t^2 x_4 - \left(1 + t^2\right) x_7\right) + p_5 \left(x_1 + t^2 x_4 - \left(1 + t^2\right) x_7\right)\right) == 0$$

In[∘]:= `Expand@Collect[lhs, (p | x)_7, F] /. F → Simplify`

Out[∘]=

$$p_7 \left(-\frac{2 x_1}{t} + 2 t (x_2 - x_4 + x_5)\right) + \frac{1}{t}\left(t^2 p_6 x_1 + p_3 x_2 - 2 p_4 x_2 + t^2 p_4 x_3 + p_2 \left(x_1 - \left(-1 + t^2\right) x_2 - t^2 (x_3 - 2 x_4)\right) - p_3 x_4 + p_4 x_4 - t^2 p_4 x_4 + t^2 p_5 x_4 - p_4 x_5 + p_5 x_5 - t^2 p_5 x_5 - t^2 p_6 x_5 + p_5 x_6 - p_1 \left(\left(-1 + t^2\right) x_1 + t^2 x_2 + x_6\right)\right) + \frac{2 \left(t^2 p_1 - p_2 + p_4 - p_5\right) x_7}{t} + \left(\frac{2}{t} - 2 t\right) p_7 x_7$$

In[∘]:= $\int \mathbb{E}[lhs]\, d\{p_7, x_7\}$

$$» \begin{pmatrix} 0 & -\frac{2-2 t^2}{t} \\ -\frac{2-2 t^2}{t} & 0 \end{pmatrix}$$

Out[∘]=

$$-\frac{1}{2\left(-1 + t^2\right)} \mathbb{i}\, t\, \mathbb{E}\left[-\frac{\left(1 + t^4\right) p_1 x_1}{(-1 + t)\, t\, (1 + t)} + \frac{\left(1 + t^2\right) p_2 x_1}{(-1 + t)\, t\, (1 + t)} - \frac{2 p_4 x_1}{(-1 + t)\, t\, (1 + t)} + \frac{2 p_5 x_1}{(-1 + t)\, t\, (1 + t)} + t p_6 x_1 + \frac{t\left(1 + t^2\right) p_1 x_2}{(-1 + t)(1 + t)} - \frac{\left(1 + t^4\right) p_2 x_2}{(-1 + t)\, t\, (1 + t)} + \frac{p_3 x_2}{t} + \frac{2 p_4 x_2}{(-1 + t)\, t\, (1 + t)} - \frac{2 t p_5 x_2}{(-1 + t)(1 + t)} - t p_2 x_3 + t p_4 x_3 - \frac{2 t^3 p_1 x_4}{(-1 + t)(1 + t)} + \frac{2 t^3 p_2 x_4}{(-1 + t)(1 + t)} - \frac{p_3 x_4}{t} - \frac{\left(1 + t^4\right) p_4 x_4}{(-1 + t)\, t\, (1 + t)} + \frac{t\left(1 + t^2\right) p_5 x_4}{(-1 + t)(1 + t)} + \frac{2 t^3 p_1 x_5}{(-1 + t)(1 + t)} - \frac{2 t p_2 x_5}{(-1 + t)(1 + t)} + \frac{\left(1 + t^2\right) p_4 x_5}{(-1 + t)\, t\, (1 + t)} - \frac{\left(1 + t^4\right) p_5 x_5}{(-1 + t)\, t\, (1 + t)} - t p_6 x_5 - \frac{p_1 x_6}{t} + \frac{p_5 x_6}{t}\right]$$

In[∘]:= `Expand@Collect[rhs, (p | x)_7, F] /. F → Simplify`

Out[∘]=

$$\frac{2 p_7 \left(x_1 - x_2 + t^2 x_4 - x_5\right)}{t} + \frac{1}{t}\left(-2 p_5 x_1 + t^2 p_6 x_1 + p_3 x_2 + t^2 p_4 x_3 - p_2 \left(x_1 + \left(-1 + t^2\right) x_2 + t^2 x_3\right) - p_3 x_4 + p_4 x_4 - t^2 p_4 x_4 - t^2 p_5 x_4 + p_4 x_5 + p_5 x_5 - t^2 p_5 x_5 - t^2 p_6 x_5 + p_1 \left(-\left(\left(-1 + t^2\right) x_1\right) + t^2 x_2 + 2 t^2 x_5 - x_6\right) + p_5 x_6\right) + \left(-2 t p_1 + 2 t p_2 - \frac{2 p_4}{t} + 2 t p_5\right) x_7 + \left(\frac{2}{t} - 2 t\right) p_7 x_7$$

In[ ]:= $\left(\int \mathbb{E}[\text{rhs}]\, d\{p_7, x_7\}\right) == \left(\int \mathbb{E}[\text{lhs}]\, d\{p_7, x_7\}\right)$

Out[ ]=

True

R1s



In[ ]:= **R1s = Simplify@**$\{Q_{2,3,2,1}, \overline{Q}_{2,3,2,1}, \overline{Q}_{1,2,1,3}, Q_{1,2,1,3}\}$

Out[ ]=

$\{0, 0, 0, 0\}$

R2C$_+$



In[ ]:= **lhs = Simplify**$[\overline{Q}_{4,1,2,5} + Q_{2,3,4,5}]$

Out[ ]=

$$t\, p_2\, (x_1 - x_3) + t\, p_4\, (-x_1 + x_3) - \frac{(p_1 - p_3)\, (x_2 - x_4)}{t}$$

In[ ]:= **Expand@Collect[lhs, (p | x)$_5$, F] /. F → Simplify**

Out[ ]=

$$t\, p_2\, (x_1 - x_3) + t\, p_4\, (-x_1 + x_3) - \frac{(p_1 - p_3)\, (x_2 - x_4)}{t}$$

In[ ]:= **Simplify[Expand@Collect[lhs, (p | x)$_5$, F] /. F → Simplify /. (p | x)$_3$ → 0]**

Out[ ]=

$$t\, p_2\, x_1 - t\, p_4\, x_1 + \frac{p_1\, (-x_2 + x_4)}{t}$$

In[ ]:= **Simplify[Expand@Collect[lhs, (p | x)$_5$, F] /. F → Simplify /. (p | x)$_4$ → 0]**

Out[ ]=

$$\frac{(-p_1 + p_3)\, x_2}{t} + t\, p_2\, (x_1 - x_3)$$